APPENDIX 3. ELECTRON DENSITY SUBROUTINES WITH INPUT PARAMETER FORMS

The following electron density models are available. The input parameter forms, which describe the model, and the subroutine listings are given on the pages shown.

A further source of versatility in this ray tracing program is the ease with which specific ionospheric models, suited to the users needs, may be introduced. To add electron density models not included in the program, the user must write a subroutine that calculates the normalized electron density (X) and its gradient ($\partial X/\partial r$, $\partial X/\partial \theta$, $\partial X/\partial \varphi$) as a function of position in spherical coordinates (r, θ, φ). ($X = 80.5 \times 10^{-6} N/f^2$, where N is the electron density in cm^{-3} and f is the wave frequency in MHz.)

Both X and its gradient must be continuous functions of position. The formulas for $\partial X/\partial r$, $\partial X/\partial \theta$, and $\partial X/\partial \varphi$ must be consistent with the variation of X with r, θ, and φ. Otherwise, the program will run slowly and give incorrect results.

The coordinates r, θ, φ refer to the computational coordinate system, which may not be the same as geographic coordinates. In particular, they are geomagnetic coordinates when the earth-centered dipole model of the earth's magnetic field is used.

The input to the subroutine (r, θ, ϕ) is through blank common. (See

Table 3.) The output is through common block /XX/. (See Table 8.) It is useful if the name of the subroutine suggests the model to which it corresponds. The subroutine should have an entry point ELECTX so that other subroutines in the program can call it. Any parameters needed by the subroutine should be input into W101 through W149 of the W array. (See Table 2.) If the model needs massive amounts of data, these should be read in by the subroutine following the example of TABLEX. As in the already existing electron density subroutines, provision should be made for perturbations to the electron density model (irregularities) by having the statement

```
IF(PERT.NE.0.) CALL ELECT1
```

before the RETURN statement at the end of the subroutine.

INPUT PARAMETER FORM FOR SUBROUTINE TABLEX

IONOSPHERIC ELECTRON DENSITY PROFILE

First card tells how many profile points in I4 format. The cards following the first card give the height and electron density of the profile points one point per card in F8. 2, E12. 4 format. The heights must be in increasing order. Set W100 = 1. 0 to read in a new profile. After the cards are read, TABLEX will reset W100 = 0. 0. This subroutine makes an exponential extrapolation down using the bottom 2 points in the profile.

1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20	1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20
HEIGHT h km	ELECTRON DENSITY N ELECTRONS/cm^3	HEIGHT h km	ELECTRON DENSITY N ELECTRONS/cm^3

```
      SUBROUTINE TABLEX
C         CALCULATES ELECTRON DENSITY AND GRADIENT FROM PROFILES HAVING
C         THE SAME FORM AS THOSE USED BY CROFTS RAY TRACING PROGRAM
C MAKES AN EXPONENTIAL EXTRAPOLATION DOWN USING THE BOTTOM TWO POINTS
C         NEEDS SUBROUTINE GAUSEL
      DIMENSION HPC(250),FN2C(250),ALPHA(250),BETA(250),GAMMA(250),
     1 DELTA(250),SLOPE(250),MAT(4,5)
      COMMON /CONST/ PI,PIT2,PID2,DEGS,RAD,K,DUM(2)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(READFN,W(100)),(PERT,W(150))
      REAL MAT,K
      DATA (MODX(1)=6HTABLEX)
      ENTRY ELECTX
      IF (READFN.EQ.0.) GO TO 10
      READFN=0.
      READ 1000, NOC,(HPC(I),FN2C(I),I=1,NOC)
 1000 FORMAT (I4/(F8.2,E12.4))
      PRINT 1200, (HPC(I),FN2C(I),I=1,NOC)
 1200 FORMAT(1H1,14X,6HHEIGHT,4X,16HELECTRON DENSITY/(1X,F20.10,E20.10))
      A=0.
      IF(FN2C(1).NE.0.) A=ALOG(FN2C(2)/FN2C(1))/(HPC(2)-HPC(1))
      FN2C(1)=K*FN2C(1)
      FN2C(2)=K*FN2C(2)
      SLOPE(1)=A*FN2C(1)
      SLOPE(NOC)=0.
      NMAX=1
      DO 6 I=2,NOC
      IF (FN2C(I).GT.FN2C(NMAX)) NMAX=I
      IF (I.EQ.NOC) GO TO 4
      FN2C(I+1)=K*FN2C(I+1)
      DO 3 J=1,3
      M=I+J-2
      MAT(J,1)=1.
      MAT(J,2)=HPC(M)
      MAT(J,3)=HPC(M)**2
    3 MAT(J,4)=FN2C(M)
      CALL GAUSEL (MAT,4,3,4,NRANK)
      IF (NRANK.LT.3) GO TO 60
      SLOPE(I)=MAT(2,4)+2.*MAT(3,4)*HPC(I)
    4 DO 5 J=1,2
      M=I+J-2
      MAT(J,1)=1.
      MAT(J,2)=HPC(M)
      MAT(J,3)=HPC(M)**2
      MAT(J,4)=HPC(M)**3
      MAT(J,5)=FN2C(M)
      L=J+2
      MAT(L,1)=0.
      MAT(L,2)=1.
      MAT(L,3)=2.*HPC(M)
      MAT(L,4)=3.*HPC(M)**2
    5 MAT(L,5)=SLOPE(M)
      CALL GAUSEL (MAT,4,4,5,NRANK)
      IF (NRANK.LT.4) GO TO 60
      ALPHA(I)=MAT(1,5)
      BETA(I)=MAT(2,5)
      GAMMA(I)=MAT(3,5)
    6 DELTA(I)=MAT(4,5)
      HMAX=HPC(NMAX)
      NH=2
   10 H=R(1)-EARTHR
      F2=F*F
      PXPR=0.
      IF (H.GE.HPC(1)) GO TO 12
   11 NH=2
      X=0.
```

```
      IF(FN2C(1).EQ.0.) GO TO 50                                        TABX068
      X=FN2C(1)*EXP(A*(H-HPC(1)))/F2                                    TABX069
      PXPR=A*X                                                          TABX070
      GO TO 50                                                          TABX071
   12 IF (H.GE.HPC(NOC)) GO TO 18                                       TABX072
      NSTEP=1                                                           TABX073
      IF (H.LT.HPC(NH-1)) NSTEP=-1                                      TABX074
   15 IF (HPC(NH-1).LE.H.AND.H.LT.HPC(NH)) GO TO 16                     TABX075
      NH=NH+NSTEP                                                       TABX076
      GO TO 15                                                          TABX077
   16 X=(ALPHA(NH)+H*(BETA(NH)+H*(GAMMA(NH)+H*DELTA(NH))))/F2           TABX078
      PXPR=(BETA(NH)+H*(2.*GAMMA(NH)+H*3.*DELTA(NH)))/F2                TABX079
      GO TO 50                                                          TABX080
   18 X=FN2C(NOC)/F2                                                    TABX081
   50 IF (PERT.NE.0.) CALL ELECT1                                       TABX082
      RETURN                                                            TABX083
   60 PRINT 6000, I,HPC(I)                                              TABX084
 6000 FORMAT(4H THE,I4,55HTH POINT IN THE ELECTRON DENSITY PROFILE HAS TTABX085
     1HE HEIGHT,F8.2,40H KM, WHICH IS THE SAME AS ANOTHER POINT.)       TABX086
      CALL EXIT                                                         TABX087
         END                                                            TABX088
```

```
      SUBROUTINE GAUSEL (C,NRD,NRR,NCC,NSF)                             GAUS001
C********* SAME AS SUBROUTINE GAUSSEL WRITTEN BY L. DAVID LEWIS ********GAUS002
      DIMENSION C(NRD,NCC),L(128,2)                                     GAUS003
C           BITS = 2.**-18                                              GAUS004
      DATA (BITS=3.8146972656E-6)                                       GAUS005
      NR=NRR                                                            GAUS006
      NC=NCC                                                            GAUS007
      IF(NC.LT.NR.OR.NR.GT.128.OR.NR.LE.0) CALL EXIT                    GAUS008
C                                                                       GAUS009
C           INITIALIZE.                                                 GAUS010
      NSF=0                                                             GAUS011
      NRM=NR-1                                                          GAUS012
      NRP=NR+1                                                          GAUS013
      D=1.                                                              GAUS014
      LSD=1                                                             GAUS015
      DO 1 KR=1,NR                                                      GAUS016
      L(KR,1)=KR                                                        GAUS017
1     L(KR,2)=0                                                         GAUS018
      IF(NR.EQ.1) GO TO 42                                              GAUS019
C                                                                       GAUS020
C           ELIMINATION PHASE.                                          GAUS021
      DO 41 KP=1,NRM                                                    GAUS022
      KPP=KP+1                                                          GAUS023
      PM=0.                                                             GAUS024
      MPN=0                                                             GAUS025
C                                                                       GAUS026
C           SEARCH COLUMN KP FROM DIAGONAL DOWN FOR MAX PIVOT.          GAUS027
      DO 2 KR=KP,NR                                                     GAUS028
      LKR=L(KR,1)                                                       GAUS029
      PT=ABS(C(LKR,KP))                                                 GAUS030
      IF(PT.LE.PM) GO TO 2                                              GAUS031
      PM=PT                                                             GAUS032
      MPN=KR                                                            GAUS033
      LMP=LKR                                                           GAUS034
2     CONTINUE                                                          GAUS035
C                                                                       GAUS036
C           IF MAX PIVOT IS ZERO, MATRIX IS SINGULAR.                   GAUS037
      IF(MPN.EQ.0) GO TO 9                                              GAUS038
      NSF=NSF+1                                                         GAUS039
      IF(MPN.EQ.KP) GO TO 3                                             GAUS040
```

```
C
C          NEW ROW NUMBER KP HAS MAX PIVOT.
      LSD=-LSD
      L(KP,2)=L(MPN,1)=L(KP,1)
      L(KP,1)=LMP
C
C          ROW OPERATIONS TO ZERO COLUMN KP BELOW DIAGONAL.
3     MKP=L(KP,1)
      P=C(MKP,KP)
      D=D*P
      DO 41 KR=KPP,NR
      MKR=L(KR,1)
      Q=C(MKR,KP)/P
      IF(Q.EQ.0.) GO TO 41
C
C          SUBTRACT Q * PIVOT ROW FROM ROW KR.
      DO 4 LC=KPP,NC
      R=Q*C(MKP,LC)
      C(MKR,LC)=C(MKR,LC)-R
4     IF(ABS(C(MKR,LC)).LT.ABS(R)*BITS) C(MKR,LC)=0.
41    CONTINUE
C
C          LOWER RIGHT HAND CORNER.
42    LNR=L(NR,1)
      P=C(LNR,NR)
      IF(P.EQ.0.) GO TO 9
      NSF=NSF+1
      D=D*P*LSD
      IF(NR.EQ.NC) GO TO 8
C
C          BACK SOLUTION PHASE.
      DO 61 MC=NRP,NC
      C(LNR,MC)=C(LNR,MC)/P
      IF(NR.EQ.1) GO TO 61
      DO 6 LL=1,NRM
      KR=NR-LL
      MR=L(KR,1)
      KRP=KR+1
      DO 5 MS=KRP,NR
      LMS=L(MS,1)
      R=C(MR,MS)*C(LMS,MC)
      C(MR,MC)=C(MR,MC)-R
5     IF(ABS(C(MR,MC)).LT.ABS(R)*BITS) C(MR,MC)=0.
6     C(MR,MC)=C(MR,MC)/C(MR,KR)
61    CONTINUE
C
C          SHUFFLE SOLUTION ROWS BACK TO NATURAL ORDER.
      DO 71 LL=1,NRM
      KR=NR-LL
      MKR=L(KR,2)
      IF(MKR.EQ.0) GO TO 71
      MKP=L(KR,1)
      DO 7 LC=NRP,NC
      Q=C(MKR,LC)
      C(MKR,LC)=C(MKP,LC)
7     C(MKP,LC)=Q
71    CONTINUE
C
C          NORMAL AND SINGULAR RETURNS.  GOOD SOLUTION COULD HAVE D=0.
8     C(1,1)=D
      GO TO 91
9     C(1,1)=0.
   91 RETURN
      END
```

INPUT PARAMETER FORM FOR SUBROUTINE CHAPX

An ionospheric electron density model consisting of a Chapman layer with tilts, ripples, and gradients

$$f_N^2 = f_c^2 \exp\left(\alpha \left(1 - z - e^{-z}\right) \right)$$

$$z = \frac{h - h_{max}}{H}$$

$$f_c^2 = f_{co}^2 \left(1 + A \sin\left(2\pi \left(\theta - \frac{\pi}{2}\right)/B \right) + C \left(\theta - \frac{\pi}{2}\right)\right)$$

$$h_{max} = h_{max_o} + E\left(\theta - \frac{\pi}{2}\right) R_o$$

f_N is the plasma frequency

h is the height above the ground

R_o is the radius of the earth in km

and θ is the colatitude in radians.

Specify:

Critical frequency at the equator, f_{c_o} = ________________ MHz (W101)

Height of the maximum electron density at the equator, h_{max_o} = ____ km (W102)

Scale height, H = ________________ km (W103)

α = ___________ (W104, 0.5 for an α Chapman layer, 1.0 for a β Chapman layer)

Amplitude of periodic variation of f_c^2 with latitude, A = ____________ (W105)

Period of variation of f_c^2 with latitude, B = _____________ rad / deg / km (W106)

Coefficient of linear variation of f_c^2 with latitude, C = _______ rad^{-1} (W107)

Tilt of the layer, E = _______________ rad / deg (W108)

```
      SUBROUTINE CHAPX
C          CHAPMAN LAYER WITH TILTS, RIPPLES, AND GRADIENTS
C          W(104) = 0.5 FOR AN ALPHA-CHAPMAN LAYER
C                 = 1.0 FOR A BETA-CHAPMAN LAYER
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (THETA,R(2))
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FC,W(101)),(HM,W(102)),
     1 (SH,W(103)),(ALPHA,W(104)),(A,W(105)),(B,W(106)),(C,W(107)),
     2 (E,W(108)),(PERT,W(150))
      DATA (MODX(1)=6H CHAPX)
      ENTRY ELECTX
      THETA2=THETA-PID2
      HMAX=HM+EARTHR*E*THETA2
      H=R(1)-EARTHR
      Z=(H-HMAX)/SH
      D=0.
      IF (B.NE.0.) D=PIT2/B
      TEMP=1.+A*SIN(D*THETA2)+C*THETA2
      EXZ=1.-EXP(-Z)
      X=(FC/F)**2*TEMP*EXP(ALPHA*(EXZ-Z))
      PXPR=-ALPHA*X*EXZ/SH
      PXPTH=X*(D*A*SIN(PID2-D*THETA2)+C)/TEMP-PXPR*EARTHR*E
      IF (PERT.NE.0.) CALL ELECT1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE VCHAPX

An ionospheric electron density model consisting of a Chapman layer with variable scale height

$$f_N^2 = f_c^2 \, \tau^{\frac{1}{2}} \, e^{\frac{1}{2}(1 - \tau)}$$

$$\tau = \left(\frac{h_{max}}{h}\right)^{\chi}$$

h is the height above the ground.

Specify:

critical frequency, fc = ___________ MHz (W101)

height of maximum electron density, h_{max} = ___________ km (W102)

χ = ___________ (W103)

```
      SUBROUTINE VCHAPX
C          CHAPMAN LAYER WITH VARIABLE SCALE HEIGHT
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FC,W(101)),(HM,W(102)),
     1 (CHI,W(103)),(PERT,W(150))
      DATA (MODX(1)=6HVCHAPX)
      ENTRY ELECTX
      HMAX=HM
      X=PXPR=0.
      H=R(1)-EARTHR
      IF (H.LE.0.) GO TO 50
      TAU=(HM/H)**CHI
      X=(FC/F)**2*SQRT(TAU)*EXP(0.5*(1.-TAU))
      PXPR=.5*X*(TAU-1.)*CHI/H
   50 IF (PERT.NE.0.) CALL ELECT1
      RETURN
      END
```

INPUT PARAMETER FORM FOR SUBROUTINE DCHAPT

An ionospheric electron density model consisting of a double, tilted α-Chapman layer

$$f_N^2 = f_{c1}^2 \exp \frac{1}{2} (1 - z_1 - e^{-z_1}) + f_{c2}^2 \exp \frac{1}{2} (1 - z_2 - e^{-z_2})$$

$$z_1 = \frac{h - h_{m1}}{H_1} \; ; \; z_2 = \frac{h - h_{m2}}{H_2}$$

$$f_{c1}^2 = f_{c10}^2 \; C(\theta - \pi/2)$$

$$f_{c2}^2 = f_{c20}^2 \; C(\theta - \pi/2)$$

$$h_{m1} = h_{m10} + R_o E \left(\frac{\pi}{180}\right) \left(\theta - \frac{\pi}{2}\right)$$

$$h_{m2} = h_{m20} + R_o E \left(\frac{\pi}{180}\right) \left(\theta - \frac{\pi}{2}\right)$$

Specify:

f_{c10} =	______ MHz	(f_{c1} at equator)	(W101)
h_{m10} =	______ Km	(h_{m1} at equator)	(W102)
H_1 =	______ Km		(W103)
f_{c20} =	______ MHz	(f_{c2} at equator)	(W104)
h_{m20} =	______ Km	(h_{m2} at equator)	(W105)
H_2 =	______ Km		(W106)
C =	______ rad^{-1}	(fractional change in f_{c1}, f_{c2}, position for increases southward)	(W107)
E =	______ deg	(positive for upward tilt to the south)	(W108)

```
      SUBROUTINE DCHAPT
C          TWO CHAPMAN LAYERS WITH TILTS
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FC1,W(101)),(HM1,W(102)),
     1 (SH1,W(103)),(FC2,W(104)),(HM2,W(105)),(SH2,W(106)),(C,W(107)),
     2 (E,W(108)),(PERT,W(150))
      DATA (MODX(1)=6HDCHAPT)
      ENTRY ELECTX
      EARTHE=EARTHR*E
      THETA2=R(2)-PID2
      HMAX=HM1+EARTHE*THETA2
      X=PXPR=PXPTH=0.
      H=R(1)-EARTHR
      IF (H.LT.0.) GO TO 50
      Z1=(H-HMAX)/SH1
      EXPZ1=1.-FXP(-Z1)
      TEMP=1.+C*THETA2
      X=(FC1/F)**2*TEMP*EXP(.5*(EXPZ1-Z1))
      PXPR=-0.5*X*EXPZ1/SH1
      PXPTH=X*C/TEMP-PXPR*EARTHE
      IF (FC2.EQ.0.) GO TO 50
      Z2=(H-HM2-EARTHE*THETA2)/SH2
      EXPZ2=1.-EXP(-Z2)
      X2=(FC2/F)**2*TEMP*EXP(.5*(EXPZ2-Z2))
      X=X+X2
      PXPR2=-0.5*X2*EXPZ2/SH2
      PXPR=PXPR+PXPR2
      PXPTH=PXPTH+X2*C/TEMP-PXPR2*EARTHE
   50 IF (PERT.NE.0.) CALL ELECT1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE LINEAR

An ionospheric electron density model consisting of a linear layer

$N = 0$ for $h \leq h_{min}$

$N = A(h - h_{min})$ for $h > h_{min}$

The ray will penetrate if $h > h_{max}$.

Specify:

$A =$ __________ electrons/cm^3/ km (W101)

$h_{max} =$ __________ km (W102)

$h_{min} =$ __________ km (W103)

```
      SUBROUTINE LINEAR
C        LINEAR ELECTRON DENSITY MODEL
      COMMON /CONST/ PI,PIT2,PID2,DEGS,RAD,K,DUM(2)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FACT,W(101)),(HM,W(102)),
     1 (HMIN,W(103)),(PERT,W(150))
      REAL K
      DATA (MODX(1)=6HLINEAR)
      ENTRY ELECTX
      H=R(1)-EARTHR
      HMAX=HM
      X=PXPR=0.
      IF (H.LE.HMIN) GO TO 50
      PXPR=K*FACT/F**2
      X=PXPR*(H-HMIN)
   50 IF (PERT.NE.0.) CALL ELECT1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE QPARAB

An ionospheric electron density model consisting of a parabolic or a quasi-parabolic layer (concentric)

$$f_N^2 = f_c^2 \left[1. - \left(\frac{h-h_{max}}{Y_m} \cdot C\right)^2\right] \quad \text{if } f_N^2 > 0.$$

$f_N^2 = 0.$ otherwise.

$C = 1.$ for a parabolic layer

$$C = \frac{R_o + h_{max} - Y_m}{R_o + h} \quad \text{for a quasi-parabolic layer}$$

where R_o is the radius of the earth.

Specify:

Critical frequency,. f_c = __________Mc/s (W101)

Height of maximum electron density, h_{max} = __________km. (W102)

Semi-thickness, Y_m = __________km. (W103)

Type of profile:

Plain parabolic______________(W104 = 0.)

Quasi-parabolic______________(W104 = 1.)

```
      SUBROUTINE QPARAB
C        PLAIN PARABOLIC OR QUASI-PARABOLIC PROFILE
C        W(104) = 0. FOR A PLAIN PARABOLIC PROFILE
C               = 1. FOR A QUASI-PARABOLIC PROFILE
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FC,W(101)),(HM,W(102)),
     1 (YM,W(103)),(QUASI,W(104)),(PERT,W(150))
      DATA (MODX(1)=6HQPARAB)
      ENTRY ELECTX
      HMAX=HM
      PXPR=0.
      H=R(1)-EARTHR
      FCF2=(FC/F)**2
      CONST=1.
      IF (QUASI.EQ.1.) CONST=(EARTHR+HM-YM)/R(1)
      Z=(H-HM)/YM*CONST
      X=MAX1F(0.,FCF2*(1.-Z*Z))
      IF (X.EQ.0.) GO TO 50
      IF (QUASI.EQ.1.) CONST=(EARTHR+HM)*(EARTHR+HM-YM)/R(1)**2
      PXPR=-2.*Z*FCF2/YM*CONST
   50 IF (PERT.NE.0.) CALL ELECT1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE BULGE

An analytic ionospheric electron density model which represents the general latitude variation of the equatorial ionosphere (afternoon, equinox, sunspot maximum) - see the center panel of figure 3.18b, page 133 of Davies (1965).

The model is an alpha Chapman layer with parameters which vary with geomagnetic latitude.

$$f_N^2 = f_c^2 \, e^{\frac{1}{2}(1-z-e^{-z})}$$

where $z = \frac{h - h_{max}}{H}$

f_N is the plasma frequency

f_c is the critical frequency

h_{max} is the height of the maximum electron density

H is the scale height

h is height

---

f_c, h_{max}, H vary with geomagnetic latitude in the following way:

if $h < 100$ km, $h_{max} = 350$ km, $f_c = 15$ Mc/s

For $h \geq 100$ km,

$h_{max} = 350$ if $\lambda \geq 24°$

$h_{max} = 430 + 80 \cos\left(\frac{180}{24}\lambda\right)$ if $\lambda < 24°$

λ is the geomagnetic latitude in degrees

$$f_c = \sqrt{50\left(\frac{\lambda}{8}\right)^2 \exp\left(2 - \left|\frac{\lambda}{8}\right|\right) + 40}$$

In all cases H is determined by the constraint that

$f_N = 2$ Mc/s at 100 km.

```
      SUBROUTINE BULGE                                                  BULG001
C        ANALYTICAL MODEL OF THE VARIATION OF THE EQUATORIAL F2 LAYER   BULG002
C        IN GEOMAGNETIC LATITUDE  (EQUATORIAL BULGE AND ANOMALY)        BULG003
C        SEE FIGURE 3.18B, PAGE 133 IN DAVIES (1965).                   BULG004
C        THIS MODEL HAS NO VARIATION IN GEOMAGNETIC LONGITUDE.          BULG005
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)                                BULG006
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX                  BULG007
      COMMON R(6) /WW/ ID(10),WO,W(400)                                 BULG008
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(PERT,W(150))                  BULG009
      DATA (MODX(1)=6H BULGE)                                           BULG010
      ENTRY ELECTX                                                      BULG011
      H=R(1)-EARTHR                                                     BULG012
      PHMPTH=PFC2PTH=0.                                                 BULG013
      HMAX=350.                                                         BULG014
      FC2=225.                                                          BULG015
      IF(H.LT.100.) GO TO 2                                             BULG016
C        EQUATORIAL BULGE                                               BULG017
      BULLAT=7.5*(PID2-R(2))                                            BULG018
      IF(ABS(BUL LAT).GE.PI)  GO TO 1                                   BULG019
      HMAX=430.+80.*COS(BULLAT)                                         BULG020
      PHMPTH=600.*SIN(BUL LAT)                                          BULG021
C        EQUATORIAL ANOMALY                                             BULG022
    1 ANMLAT=22.5*(PID2-R(2))/PI                                        BULG023
      POW=2.-ABS(ANM LAT)                                               BULG024
      FC2=50.*ANM LAT**2*EXP(   POW      ) + 40.                        BULG025
      PFC2PTH=-1125./PI*POW*ANMLAT*EXP(POW)                             BULG026
C     FORCING PLASMA FREQ AT 100 KM TO BE 2 MHZ IN ORDER TO CALCULATE SH BULG027
    2 ALPHA=2.*ALOG(FC2/4.)+1.                                          BULG028
      Z100=-ALOG(ALPHA)                                                 BULG029
      DO 3  I=1,5                                                       BULG030
    3 Z100=-ALOG(ALPHA-Z100)                                            BULG031
      SH=(100.-HMAX)/Z100                                               BULG032
      Z=(H-HMAX)/SH                                                     BULG033
      EXZ=1.-EXP(-Z)                                                    BULG034
      X=FC2*EXP(.5*(EXZ-Z))/F**2                                        BULG035
      PXPR=-0.5*X*EXZ/SH                                                BULG036
      PXPTH=-PXPR*(1.-Z/Z100)*PHMPTH+(1.-Z*EXZ/(Z100*(1.-EXP(-Z100))))  BULG037
     1       *X/FC2*PFC2PTH                                             BULG038
      IF (PERT.NE.0.) CALL ELECT1                                       BULG039
      RETURN                                                            BULG040
         END                                                            BULG041-
```

INPUT PARAMETER FORM FOR SUBROUTINE EXPX

An exponential electron density profile

$$N = N_o e^{a(h-h_o)}$$

h is the height above the ground.

Specify:

the electron density at the height h_o, N_o = ____________ cm^{-3} (W101)

the reference height, h_o = ____________ km (W102)

the exponential increase of N with height, a = __________ km^{-1} (W103)

```
      SUBROUTINE EXPX                                                  EXPX001
C        EXPONENTIAL ELECTRON DENSITY MODEL                            EXPX002
      COMMON /CONST/ PI,PIT2,PID2,DEGS,RAD,K,DUM(2)                    EXPX003
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX                 EXPX004
      COMMON R(6) /WW/ ID(10),W0,W(400)                                EXPX005
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),                              EXPX006
     1 (N0,W(101)),(H0,W(102)),(A,W(103)),(PERT,W(150))                EXPX007
      REAL N, N0 ,K                                                    EXPX008
      DATA (MODX(1)=4HEXPX),(HMAX=350.)                                EXPX009
      ENTRY ELECTX                                                     EXPX010
      H=R(1)-EARTHR                                                    EXPX011
      N =N0 * EXP(A*(H-H0))                                            EXPX012
      X=K*N/F**2                                                       EXPX013
      PXPR=A*X                                                         EXPX014
      IF (PERT.NE.0.) CALL ELECT1                                      EXPX015
      RETURN                                                           EXPX016
         END                                                           EXPX017-
```